

L a n H E P —

a package for automatic generation
of Feynman rules in gauge models

A. V. Semenov

LanHEP — a package for automatic generation of Feynman rules in gauge models.

Abstract. The LanHEP program for Feynman rules generation in momentum representation is presented. It reads the Lagrangian written in the compact form close to one used in publications. It means that Lagrangian terms can be written with summation over indices of broken symmetries and using special symbols for complicated expressions, such as covariant derivative and strength tensor for gauge fields. The output is Feynman rules in terms of physical fields and independent parameters. This output can be written in LaTeX format and in the form of CompHEP model files, which allows one to start calculations of processes in the new physical model. Although this job is rather straightforward and can be done manually, it requires careful calculations and in the modern theories with many particles and vertices can lead to errors and misprints. The program allows one to introduce into CompHEP new gauge theories as well as various anomalous terms.

E-mail: `semenov@theory.npi.msu.su`
WWW page: `http://theory.npi.msu.su/~semenov/lanhep.html`

Introduction

LanHEP has been designed as a part of the CompHEP package [1], worked out for automatic calculations in high energy physics. CompHEP allows symbolic computation of the matrix element squared of any process with up to 6 incoming and outgoing particles for a given physical model (i.e. a model defined by a set of Feynman rules as a table of vertices in the momentum representation) and then numerical calculation of cross-sections and various distributions.

Main purpose of the new option given by LanHEP program is designing of a new physical model. LanHEP program makes possible the generation of Feynman rules for propagators and vertices in momentum representation starting from the Lagrangian defined by a user in some simple format very similar to canonical coordinate representation. User should prepare a text file with description of all Lagrangian terms in the format close to the form used in standard publications. Of course, user has to describe all particles and parameters appearing in Lagrangian terms.

The main LanHEP features are:

- LanHEP expands expression and combines similar terms;
- it performs the Fourier transformation by replacing derivatives with momenta of particles;
- it writes Feynman rules in the form of four tables in CompHEP format as well as tables in LaTeX format;
- user can define the substitution rules, for example for covariant derivative;
- it is possible to define multiplets, and (if necessary) their components;
- user can write Lagrangian terms with Lorentz and multiplet indices explicitly or omit indices (all or some of them);
- LanHEP performs explicit summation over the indices in Lagrangian terms, if the corresponding components for multiplets and matrices are introduced;
- it allows the user to introduce vertices with 4 fermions and 4 colored particles (such vertices can't be introduced directly in CompHEP) by means of auxiliary field with constant propagator;
- it also can check whether the set of introduced vertices satisfies the electric charge conservation law.

1 QED by means of LanHEP

We start from a simple exercises, illustrating the main ideas and features of LanHEP. The first physical model is Quantum Electrodynamics.

```
model QED/1.
parameter ee=0.31333:'elementary electric charge'.
spinor e1/E1:(electron, mass me=0.000511).
vector A/A:(photon).
let F^mu^nu=deriv^nu*A^mu-deriv^mu*A^nu.
lterm -1/4*(F^mu^nu)**2 - 1/2*(deriv^mu*A^mu)**2.
lterm E1*(i*gamma*deriv+me)*e1.
lterm ee*E1*gamma*A*e1.
```

Figure 1: LanHEP input file for the generation of QED Feynman rules

QED Lagrangian is

$$\mathcal{L}_{QED} = -\frac{1}{4}F_{\mu\nu}F^{\mu\nu} + \bar{e}\gamma^\mu(i\partial_\mu + g_e A_\mu)e - m\bar{e}e$$

and the gauge fixing term in Feynman gauge has the form

$$\mathcal{L}_{GF} = -\frac{1}{2}(\partial_\mu A^\mu)^2.$$

Here $e(x)$ is the spinor electron-positron field, m is the electron mass, $A_\mu(x)$ is the vector photon field, $F^{\mu\nu} = \partial_\nu A^\mu - \partial_\mu A^\nu$, and g_e is the elementary electric charge.

The LanHEP input file to generate the Feynman rules for QED is shown in Fig. 1.

First of all, the input file consists of statements. Each statement begins with one of the reserved keywords and ends by the full-stop '.' symbol.

First line says that this is a model with the name `QED` and number 1. This information is supplied for CompHEP, the name `QED` will be displayed in its list of models. In CompHEP package each model is described by four files: 'varsN.mdl', 'funcN.mdl', 'prtclsN.mdl', 'lgrngN.mdl', where N is the very number specified in the `model` statement.

The `model` statement stands first in the input file. If this statement is absent, LanHEP does not generate four standard CompHEP files, just builds the model and prints diagnostic if errors are found.

Second line in the input file contains declaration of the model parameter, denoting elementary electric charge g_e as `ee`. For each parameter used in the model one should declare its numeric value and optional comment (it is also used in CompHEP menus).

The next two lines declare particles. Statement names `spinor`, `vector` correspond to the particle spin. So, we declare electron denoted by `e1` (the corresponding antiparticle name is `E1`) and photon denoted by `A` (with antiparticle name being `A`, since the antiparticle for photon is identical to particle).

After particle name we give in brackets some options. The first one is full name of the particle, used in CompHEP; the second option declares the mass of this particle.

The `let` statement in the next line declares the substitution rule for symbol `F`, which will be replaced in the further Lagrangian terms by the expression given in this statement.

Predefinite name `deriv`, reserved for the derivation $\frac{\partial}{\partial x}$, will be replaced after the Fourier transformation by the momentum of the particle multiplied by $-i$.

The rest of the lines describe Lagrangian terms. Here the reserved name `gamma` denotes Dirac's γ -matrices.

One can see that the indices are written separated with the caret symbol '^'. Note that in the last two lines we have omitted indices. It means that LanHEP restores omitted indices automatically. Really, one can type the last term in the full format:

```
lterm ee*e1^a*gamma^a^b^mu*A^mu*e1^b.
```

It corresponds to $g_e \bar{e}_a \gamma_{ab}^\mu e_b A_\mu$ with all indices written. Note that the order of objects in the monomial is important to restore indices automatically.

2 QCD

Now let us consider the case of the Quantum Chromodynamics. The Lagrangian for gluon fields reads

$$L_{YM} = -\frac{1}{4} F^{a\mu\nu} F_{\mu\nu}^a,$$

where

$$F_{\mu\nu}^a = \partial_\mu G_\nu^a - \partial_\nu G_\mu^a - g_s f^{abc} G_\mu^b G_\nu^c,$$

$G_\mu^a(x)$ is the gluon field, g_s is a strong charge and f^{abc} are purely imaginary structure constants of $SU(3)$ color group.

The quark kinetic term and its interaction with the gluon has the form

$$L_F = \bar{q}_i \gamma^\mu \partial_\mu q_i + g_s \lambda_{ij}^a \bar{q}_i \gamma^\mu q_j G_\mu^a,$$

where λ_{ij}^a are Gell-Mann matrices.

Gauge fixing terms in Feynman gauge together with the corresponding Faddeev-Popov ghost term are

$$-\frac{1}{2}(\partial_\mu G_\mu^a)^2 + ig_s f^{abc} \bar{c}^a G_\mu^b \partial^\mu c^c,$$

where (c, \bar{c}) are unphysical ghost fields.

The corresponding LanHEP input file is shown in Fig. 2.

Since QCD uses objects with color indices, one has to declare the indices of these objects. There are three types of color indices supported by LanHEP. These types are referred as `color c3` (color triplets), `color c3b` (color antitriplets), and `color c8` (color octets). One can see that `color c3` index type appears among the options in the quark `q` declaration, and the `color c8` one in the gluon `G` declaration. Antiquark `Q` has got color index of type `color c3b` as antiparticle to quark. LanHEP allows convolution of an index of type `color c3` only with another index of type `color c3b`, and two indices of type `color c8`. Of course, in Lagrangian terms each index has to be convoluted with its partner, since Lagrangian has to be scalar.

```

model QCD/2.
parameter gg=1.117:'Strong coupling'.
spinor q/Q:(quark, mass mq=0.01, color c3).
vector G/G:(gluon, color c8, gauge).
let F^mu^nu^a = deriv^nu*G^mu^a - deriv^mu*G^nu^a -
      gg*f_SU3^a^b^c*G^mu^b*G^nu^c.
lterm -F**2/4-(deriv*G)**2/2.
lterm Q*(i*gamma*deriv+mq)*q.
lterm i*gg*f_SU3*ccghost(G)*G*deriv*ghost(G).
lterm gg*Q*gamma*lambda*G*q.

```

Figure 2: Input file for the generation of QCD Feynman rules

Table 1: QCD Feynman rules generated by LanHEP in LaTeX output format

Fields in the vertex				Variational derivative of Lagrangian by fields
$G_{\mu p}$	$G.C_q$	$G.C_r$		$-gg \cdot p_3^\mu f_{pqr}$
Q_{ap}	q_{bq}	$G_{\mu r}$		$gg \cdot \gamma_{ab}^\mu \lambda_{pq}^r$
$G_{\mu p}$	$G_{\nu q}$	$G_{\rho r}$		$gg f_{pqr} (p_3^\nu g^{\mu\rho} - p_2^\rho g^{\mu\nu} - p_3^\mu g^{\nu\rho} + p_1^\rho g^{\mu\nu} + p_2^\mu g^{\nu\rho} - p_1^\nu g^{\mu\rho})$
$G_{\mu p}$	$G_{\nu q}$	$G_{\rho r}$	$G_{\sigma s}$	$gg^2 (g^{\mu\rho} g^{\nu\sigma} f_{pqt} f_{rst} - g^{\mu\sigma} g^{\nu\rho} f_{pqt} f_{rst} + g^{\mu\nu} g^{\rho\sigma} f_{prt} f_{qst} + g^{\mu\nu} g^{\rho\sigma} f_{pst} f_{qrt} - g^{\mu\sigma} g^{\nu\rho} f_{prt} f_{qst} - g^{\mu\rho} g^{\nu\sigma} f_{pst} f_{qrt})$

LanHEP allows also to use in the Lagrangian terms a predefined symbol `lambda` with the three indices of types `color c3`, `color c3b`, `color c8` corresponding to Gell-Mann λ -matrices. Symbol `f_SU3` denotes the structure constant f^{abc} of color $SU(3)$ group (all three indices have the type `color c8`).

Option `gauge` in the declaration of `G` allows to use names `ghost(G)` and `ccghost(G)` for the ghost fields c and \bar{c} in Lagrangian terms and in `let` statements.

Table 1 shows Feynman rules generated by LanHEP in LaTeX format after processing the input file presented in Fig. 2. Four gluon vertex rule is indicated in the last line. Note that the output in CompHEP format has no 4-gluon vertex explicitly; it is expressed effectively through 3-leg vertices by the constant propagator of some auxiliary field (see Appendix A for more details).

3 Syntax

The LanHEP input file is the sequence of statements, each starts with a special identifier (such as `parameter`, `lterm` etc) and ends with the full-stop `'.'` symbol. Statement can occupy several lines in the input file.

This section is aimed to clarify the syntax of LanHEP input files, i.e. the structure of the statements.

3.1 Constants and identifiers

First of all, each word in any statement is either an *identifier* or a *constant*.

Identifiers are the names of particles, parameters etc. Examples of identifiers from the previous section are particle names

```
e1 E1 A q Q G
```

The first word in each statement is also an identifier, defining the function which this statement performs. The identifiers are usually combinations of letters and digits starting with a letter. If an identifier doesn't respect this rule, it should be quoted. For example, the names of W^\pm bosons must be written as `'W+'` and `'W-'`, since they contain `'+'` and `'-'` symbols.

Constants can be classified as

- *integers*: they consist of optional sign followed by one or more decimal digits, such as

```
0 1 -1 123 -98765
```

Integers can appear in Lagrangian terms, parameter definition and in other expressions.

- *Floating point numbers* include optional sign, several decimal digits of mantissa with an embedded period (decimal point) with at least one digit before and after the period, and optional exponent. The exponent, if present, consist of letter E or e followed by an optional sign and one or more decimal digits. The valid examples of floating point numbers are

```
1.0  -1.0  0.000511  5.11e-4
```

Floating point numbers are used only as parameter values (coupling constants, particle masses etc). They can not be explicitly used in Lagrangian terms.

- *String constants* may include arbitrary symbols. They are used as comments in parameter statements, full particle names in the declaration of a particle, etc. Examples from the previous section are

```
electron
photon
```

If a string constant contains any character besides letters and digits or doesn't begin with a letter, it should be quoted. For example, the comments in QED and QCD input files (see previous section) contain blank spaces, so they are quoted:

```
'elementary electric charge'
'Strong coupling'
```

3.2 Comments

User can include comments into the LanHEP input file in two ways. First, symbol '%' denotes the comment till the end of current line. Second way allows one to comment any number of lines by putting a part of input file between '/*' (begin of comment) and '*/' (end of comment) symbols.

4 Objects in the expressions for Lagrangian terms

Each symbol which may appear in algebraic expressions (names of parameters, fields, etc) has a fixed order of indices and their types. If this object is used in any expression, one should write its indices in the same order as they were defined when the object has been declared.

Besides the indices types corresponding to color $SU(3)$ group: `color c3` (color triplet), `color c3b` (color antitriplet) and `color c8` (color octet) described in the previous example, there are default types of indices for Lorentz group: `vector`, `spinor` and `cspinor` (antispinor). User can also declare new types of indices corresponding to the symmetries other than color $SU(3)$ group. In this case any object (say, particle) may have indices related to this new group. This possibility will be described in Section 7.

If an index appears twice in some monomial of an expression, LanHEP assumes summation over this index. Types of such indices must allow the convolution, i.e. they should be one of the pairs: `spinor` and `cspinor`, two `vector`, `color c3` and `color c3b`, two `color c8`.

In general the following objects are available to appear in the expressions for a Lagrangian: integers and identifiers of parameters, particles, specials, let-substitutions and arrays.

There are also predefined symbols `i`, denoting imaginary unit i ($i^2 = -1$) and `Sqrt2`, which is a parameter with value the $\sqrt{2}$.

4.1 Parameters

Parameters are scalar objects (i.e. they have no indices). Parameters denote coupling constants, masses and widths of particles, etc. To introduce a new parameter one should use the `parameter` statement, which has the generic form

```
parameter name=value:comment.
```

- *name* is an identifier of newly created parameter.
- *value* is an integer or floating point number or an expression. One can use previously declared parameters and integers joined by standard arithmetical operators '+', '-', '*', '/', and '**' (power).

- comment is an optional comment to clarify the meaning of parameter, it is used in CompHEP help windows. Comment has to be a string constant, so if it contains blank spaces or other special characters, it must be quoted (see Section 3).

4.2 Particles

Particles are objects to denote physical particles. They may have indices. It is possible to use three statements to declare a new particle, at the same time the second and the third statements define the corresponding Lorentz index:

```
scalar P/aP:(options) .
spinor P/aP:(options) .
vector P/aP:(options) .
```

P and aP are identifiers of particle and antiparticle. In the case of truly neutral particles (when antiparticle is identical to the particle itself) one should use the form P/P with identical names for particle and antiparticle.

It is possible to write only the particle name, e.g.

```
scalar P:(options) .
```

In this case the name of corresponding antiparticle is generated automatically. It satisfies the usual CompHEP convention, when the name of antiparticle differs from particle by altering the case of the first letter. So for electron name **e1** automatically generated antiparticle name will be **E1**. If the name contains symbol '+' it is replaced by '-' and vice versa.

The *option* is comma-separated list of options for a declared particle, and it may include the following items:

- the first element in this list must be the full name of the particle, (e.g. **electron** and **photon** in our example.) Full name is string constant, so it should be quoted if it contains blanks, etc.
- mass *param=value* defines the mass of the particle. Here *param* is an identifier of new parameter, which is used to denote the mass; *value* is its value, it has the same syntax as in the **parameter** statement, comment for this new parameter being generated automatically. If this option omitted, the mass is assumed to be zero.
- width *param=value* declares the width of the particle. It has the same syntax as for **mass** option.
- itype is a type of index of some symmetry; one can use default index types for color $SU(3)$ group (see QCD example in Section 2). It is possible to use user-defined index types (see Section 7) and Lorentz group indices **vector**, **spinor**, **cspinor**.
- left or right say that the massless spinor particle is an eigenstate of $(1 - \gamma^5)/2$ or $(1 + \gamma^5)/2$ projectors, so this fermion is left-handed or right-handed one.
- gauge declares the vector particle as a gauge boson. This option generates corresponding ghosts and goldstone bosons names for the named particle (see below).

When a particle name is used in any expression (in Lagrangian terms), one should remember that the first index is either vector or spinor one (of course, if this particle is not a Lorentz scalar). Then the indices follow in the same order as index types in the *options* list. So, in the case of quark declaration (see QCD example) the first index is spinor, and the second one is color triplet.

There are several functions taking particle name as an argument which can be used in algebraic expressions. These functions are replaced with auxiliary particle names, which are generated automatically.

- Ghost field names in gauge theories are generated by the functions **ghost**(*name*) \rightarrow '*name.c*' and **ccghost**(*name*) \rightarrow '*name.C*' (see for instance Table 1). Here and below *name* is the name of the corresponding gauge boson.
- Goldstone boson field name in the t'Hooft-Feynman gauge is generated by the function **gsb**(*name*) \rightarrow '*name.f*'.
- The function **anti**(*name*) generates antiparticle name for the particle *name*.
- The name for a charge conjugated spinor particle $\psi^c = C\bar{\psi}^T$ is generated by the function **cc**(*name*) \rightarrow '*name.c*'. Charge conjugated fermion has the same indices types and ordering, however index of **spinor** type is replaced by the index type **cspinor** and vice versa.

- `vev(expr)` is used in Lagrangian terms for vacuum expectation values. Function `vev` ensures that `deriv*vev(expr)` is zero. In other words, `vev` function forces LanHEP to treat `expr` as a scalar particle which will be replaced by `expr` in Feynman rules.

4.3 Specials

Besides parameters and particles other indexed such as γ -matrices, group structure constants, etc may appear in the Lagrangian terms. We refer such objects as *specials*.

Predefined specials of Lorentz group are:

- `gamma` stands for the γ -matrices. It has three indices of `spinor`, `cspinor` and `vector` types.
- `gamma5` denotes γ^5 matrix. It has two indices of `spinor` and `cspinor` types.
- `moment` has one index of `vector` type. At the stage of Feynman rules generation this symbol is replaced by the particle moment.
- `deriv` is replaced by $-ip_\mu$, where p_μ is the particle moment. It has one `vector` index.

Specials of color $SU(3)$ group are:

- `lambda` denotes Gell-Mann λ -matrices. It has three indices: `color c3`, `color c3b` and `color c8`.
- `f_SU3` is the $SU(3)$ structure constant. It has three indices of `color c8` type.

Note that for specials the order of indices types is fixed.

User can declare new specials with the help of a facility to introduce user-defined indices types (see Section 7).

4.4 Let-substitutions

LanHEP allows the user to introduce new symbols and then substitute them in Lagrangian terms by some expressions. Substitution has the generic form

```
let name=expr.
```

where `name` is the identifier of newly defined object. The expression has the same structure as those in Lagrangian terms, however here expression may have free (non-convoluted) indices.

Typical example of using a substitution rule is a definition of the QED covariant derivative as

```
let Deriv^mu=deriv^mu + i*ee*A^mu.
```

corresponding to $D_\mu = \partial_\mu + ig_e A_\mu$.

More complicated example is the declaration $\sigma^{\mu\nu} \equiv i(\gamma^\mu\gamma^\nu - \gamma^\nu\gamma^\mu)/2$ matrices:

```
let sigma^a^b^mu^nu = i*(gamma^a^c^mu*gamma^c^b^nu - gamma^a^c^nu*gamma^c^b^mu)/2.
```

Note that the order of indices types of new symbol is fixed by the declaration. So, first two indices of `sigma` after this declaration are `spinor` and `antispinor`, third and fourth are `vector` indices.

4.5 Arrays

LanHEP allows to define components of indexed objects. In this case, convolution of indices will be performed as an explicit sum of products of the corresponding components.

Object with explicit components has to be written as

```
{expr1, expr2 ..., exprN }^i
```

where expressions correspond to components. All indices of components (if they present) have to be written at each component, and the index numbering components has to be written after closing curly bracket. Of course, all the components must have the same types of free (non-convoluted) indices.

Arrays are usually applied for the definition of multiplets and matrices corresponding to broken symmetries.

Typical example of arrays usage is a declaration of electron-neutrino isospin doublet `l1` (and antidoublet `L1`)

```
let l1^a^I = { n1^a, e1^a }^I, L1^a^I = { N1^a, E1^a }^I.
```

Here we suppose that `n1` was declared as the spinor particle (neutrino), with the antiparticle name `N1`.

Matrices can be represented as arrays which have other arrays as components. However, it is more convenient to declare them with omitted indices, see Section 6.4 (the same is correct for multiplets also).

It is possible also to use arrays directly in the Lagrangian terms, rather than only in the declaration of let-substitution.

5 Structure of expressions for the Lagrangian

When all parameters and particles necessary for introduction of physical model are declared, one can enter Lagrangian terms with the help of the `lterm` statement:

```
lterm expr.
```

Elementary objects of expression are integers, identifiers of parameters, particles, specials, let-substitutions, and arrays.

These elementary objects can be combined by usual arithmetical operators as

- $expr1+expr2$ (addition),
- $expr1-expr2$ (subtraction),
- $expr1*expr2$ (product),
- $expr1/expr2$ (fraction; here $expr2$ must be a product of integers and parameters),
- $expr1**N$ (Nth power of $expr1$; N must be integer).

One can use brackets '(' and ')' to force the precedence of operators. Note, that indices can follow only elementary objects symbols, i.e. if $A1$ and $A2$ were declared as two vector particles then valid expression for their sum is $A1^\mu+A2^\mu$, rather than $(A1+A2)^\mu$.

5.1 Where-substitutions

More general form of expressions involves *where-substitutions*:

```
expr where subst.
```

In the simple form *subst* is *name=repl* or several constructions of such kind separated by comma ','. In the form of such kind each instance of identifier *name* in *expr* is replaced by *repl*.

Note that in contrast to let-substitutions, where-substitution doesn't create a new object. LanHEP simply replaces *name* by *repl*, and then processes the resulting expression. It means in particular that *name* can not have indices, although it can denote an object with indices:

```
lterm F**2 where F=deriv^mu*A^nu-deriv^nu*A^mu.
```

is equivalent to

```
lterm (deriv^mu*A^nu-deriv^nu*A^mu)**2.
```

The substitution rule introduced by the keyword **where** is active only within the current **lterm** statement.

More general form of where-substitution allows to use several *name=repl* substitution rules separated by semicolon ';'. In this case *expr* will be replaced by the sum of expressions; each term in this sum is produced by applying one of the substitution rules from semicolon-separated list to the expression *expr*. This form is useful for writing the Lagrangian where many particles have similar interaction.

For example, if u, d, s, c, b, t are declared as quark names, their interaction with the gluon may read as

```
lterm gg*anti(psi)*gamma*lambda*G*psi where
      psi=u; psi=d; psi=s; psi=c; psi=b; psi=t.
```

The equivalent form is

```
lterm gg*U*gamma*lambda*G*u + gg*D*gamma*lambda*G*d +
      gg*U*gamma*lambda*G*u + gg*D*gamma*lambda*G*d +
      gg*U*gamma*lambda*G*u + gg*D*gamma*lambda*G*d.
```

Where-substitution can also be used in **let** statement. In this case one should use brackets:

```
let lsub=(expr where wsub=expr1).
```

Note that in the previous example we have omitted indices. In the next section we shall describe this LanHEP option in details.

6 Omitting indices

Physicists usually do not write all possible indices in the Lagrangian terms. LanHEP also allows a user to omit indices. This feature can simplify introduction of expressions and make them more readable. Compare two possible forms:

```
lterm E1^a*gamma^a^b^mu*A^mu*e1^b.
```

corresponding to $g_e \bar{e}_a(x) \gamma_{ab}^\mu e_b(x) A_\mu(x)$, and

```
lterm E1*gamma^mu*e1*A^mu.
```

corresponding to $g_e \bar{e}(x) \gamma^\mu e(x) A_\mu(x)$. Furthermore, while physicists usually write vector indices explicitly in the formulas, in LanHEP vector indices also can be omitted:

```
lterm -i*ee*E1*gamma*e1*A.
```

Generally speaking, when user omits indices in the expressions, LanHEP faces two problems: which indices were omitted and how to convolute restored indices.

6.1 Restoring the omitted indices

When the indexed object is declared the corresponding set of indices is assumed. Thus, if the quark **q** is declared as

```
spinor q:('some quark', color c3).
```

its first index is spinor and the second one belongs to the **color c3** type. If both indices are omitted in some expression, LanHEP generates them in the correspondence to order (**spinor**, **color c3**). However, if only one index is written, for example in the form q^a , LanHEP has to recognize whether the index **a** is of **color c3** or of **spinor** types.

To solve this problem LanHEP looks up the *list of indices omitting order*. By default this list is set to

```
[spinor, color c3, color c8, vector]
```

The algorithm to restore omitted indices is the following. First, LanHEP assumes that user has omitted indices which belong to the first type (and corresponding antitype) from this list. Continuing the consideration of our example with particle **q** one can see that since this particle is declared having one **spinor** index (the first type in the list) LanHEP checks whether the number of indices declared for this object without **spinor** index equals to the number of indices written explicitly by user. In our example (when user has written q^a) this is true. In the following LanHEP concludes that the user omitted **spinor** index and that explicitly written index is of **color c3** type.

In other cases, when the supposition fails if the user has omitted indices of the first type in the *list of indices omitting order*, LanHEP goes to the second step. It assumes that user has omitted indices of first two types from this list. If this assumption also fails, LanHEP assumes that user has omitted indices of first three types in the list and so on. At each step LanHEP subtracts the number of indices of these types assumed to be omitted from the full number of indices declared for the object, and checks whether this number of resting indices equals to the number of explicitly written indices. If LanHEP fails when the *list of indices omitting order* is completed, error message is returned by the program.

Note that if the user would like to omit indices of some type, he must omit all indices of this type (and antitype) as well as the indices of all types which precede in the *list of indices omitting order*.

For example, if object **Y** is declared with one **spinor**, two **vector** and three **color c8** indices, than

- the form $Y^a^b^c^d^e^f$ means that the user wrote all the indices explicitly;
- the form $Y^a^b^c^d^e$ means that the user omitted **spinor** index and wrote **vector** and **color c8** ones;
- the form Y^a^b means that the user omitted **spinor** and **color c8** indices and wrote only two **vector** ones;
- the form Y means that the user omitted all indices;
- all other forms, involving different number of written indices, are incorrect.

One can say that indices should be omitted in the direct correspondence with abrupting the *list of indices omitting order* from left to right.

One could change the *list of indices omitting order types* by the statement `SetDefIndex`. For example, for default setting it looks like

```
SetDefIndex(spinor, color c3, color c8, vector).
```

Each argument in the list is a type of index.

6.2 Convolution of restored indices

Omitted index can be convoluted only with some another omitted index. LanHEP expands the expression and restore indices in each monomial. LanHEP reads objects in the monomial from the left to the right and checks whether restored indices are present. If such index appears LanHEP seeks for the restored index of the appropriate type at the next objects. Note, that the program does not check whether the object with the first restored index has another restored index of the appropriate type. Thus, if `F` is declared as let-substitution for the strength tensor of electromagnetic field (with two vector indices) then expression `F**F` (as well as `F**2`) after processing omitted indices turns to implied form $F^{\mu\nu}F^{\mu\nu}$ rather than $F^{\mu\mu}F^{\nu\nu}$.

This algorithm makes the convolutions to be sensitive to the order of objects in the monomial. Let us look again to the QED example. Expression `E1*gamma*A*e1` (as well as `A*e1*gamma*E1`) leads to correct result where vector index of photon is convoluted with the same index of γ -matrix, spinor index of electron is convoluted with antispinor index of γ -matrix and antispinor index of positron is convoluted with spinor index of γ -matrix. However the expression `e1*E1*A*gamma` leads to wrong form $e1^a E1^a A^{\mu} \gamma^b c^{\mu}$, because the first antispinor index after electron belongs to positron. Spinor indices of `gamma` stay free (non-convoluted) since no more objects with appropriate indices (so, LanHEP will report an error since Lagrangian term is not a scalar).

Note, that in the vertex with two γ -matrices the situation is more ambiguous. Let's look at the term corresponding to the electron anomalous magnetic moment $\bar{e}(x)(\gamma^{\mu}\gamma^{\nu} - \gamma^{\nu}\gamma^{\mu})e(x)F_{\mu\nu}$. The correct LanHEP expression is

```
e1*(gamma^mu*gamma^nu - gamma^nu*gamma^mu)*E1*F^mu^nu
```

Here vector indices can't be omitted, since it lead to the convolution of vector indices of γ -matrices. One can see also that the form

```
e1*E1*(gamma^mu*gamma^nu - gamma^nu*gamma^mu)*F^mu^nu
```

will correspond to the expression $\bar{e}(x)e(x)Sp(\gamma^{\mu}\gamma^{\nu})F_{\mu\nu}$. Here LanHEP has got scalar Lorentz-invariant expression in the Lagrangian term, so it has no reason to report error.

These examples mean that user should clearly realize how the indices will be restored and convoluted, or he has to write all indices explicitly.

6.3 Let-substitutions

Other problem arises when the omitted indices stay free, it is a case for the `let` statement. LanHEP allows only two ways to avoid ambiguity in the order of indices types: either user specifies all the indices at the name of new symbol and free indices in the corresponding expression, or he should omit all free indices. In the latter case the order of indices types is defined by the order of free omitted indices in the first monomial of the expression. For example if `A1` and `A2` are vectors and `c1` and `c2` are spinors, the statement

```
let d=A1*c1+c2*A2.
```

declares new object `d` with two indices, the first is vector index and the second is spinor index according to their order in the monomial `A1*c1`. Of course, each monomial in the expression must have the same typeset of free indices.

6.4 Arrays

The usage of arrays with omitted indices allows us to define matrices conveniently. For example, the declaration of τ -matrices

$$\tau_1 = \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}, \quad \tau_2 = \begin{pmatrix} 0 & i \\ -i & 0 \end{pmatrix}, \quad \tau_3 = \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}$$

can be written as

```

let tau1 = {{0, 1}, {1, 0}}.
let tau2 = {{0,-i}, {i, 0}}.
let tau3 = {{1, 0}, {0,-1}}.

```

One can see that in such way of declaration a matrix is written "column by column".

The declaration of objects with three 'explicit' indices can be done using the objects already defined. For example, when τ -matrices are defined as before, it is easy to define the vector $\vec{\tau} \equiv (\tau_1, \tau_2, \tau_3)$ as

```
let tau = {tau1, tau2, tau3}.
```

The object `tau` has three indices, first pair selects the element of the matrix, while the matrix itself is selected by the third index, i.e. `tau~i~j~a` corresponds to τ_{ij}^a .

On the other hand, the declaration of structure constants of a group is more complicated. Declaring such an object one should bear in mind that omitting indices implies that in a sequence of components the second index of an object is changed after the full cycle of the first index, the third index is changed after the full cycle of the second one, etc. For example, a declaration of the antisymmetrical tensor ε^{abc} can read as

```

let eps = {{{0,0,0}, {0,0,-1}, {0,1,0}},
            {{0,0,1}, {0,0,0}, {-1,0,0}},
            {{0,-1,0}, {1,0,0}, {0,0,0}}}.

```

One can easily see that the components are listed here in the following order:

$\varepsilon^{111}, \varepsilon^{211}, \varepsilon^{311}, \varepsilon^{121}, \varepsilon^{221}, \varepsilon^{321}, \varepsilon^{131}, \dots, \varepsilon^{233}, \varepsilon^{333}$.

The declaration of more complex objects such as $SU(3)$ structure constants can be made in the same way.

7 Declaration of new index types and indexed objects

7.1 Declaring new groups

Index type is defined by two keywords¹: *group name* and *representation name*. Thus, color triplet index type `color c3` has group name `color` and representation name `c3`.

LanHEP allows user to introduce new group names by the `group` statement:

```
group gname.
```

Here *gname* is a string constant, which becomes the name of newly declared group.

Representation names for each group name must be declared by the statement

```
repres gname:(rlist)
```

where *rlist* is a comma-separated list of representation names declaration for the already declared group name *gname*. Each such declaration has the form either *rname* or *rname/crname*. In the first case index which belongs to the *gname rname* type can be convoluted with another index of the same type; in the second case index of *gname rname* type can be convoluted only with an index of *gname crname* type.

For example, definition for color $SU(3)$ group with fundamental, conjugated fundamental and adjoint representations looks as:

```

group color:SU(3).
repres color:(c3/c3b,c8).

```

So, three index types can be used: `color c3`, `color c3b`, `color c8`. The convolution of these indices is allowed by pairs (`color c3`, `color c3b`) and (`color c8`, `color c8`) indices.

7.2 Declaring new specials

Specials with indices of user-defined types can be declared by means of `special` statement:

```
special name:(ilist).
```

Here *name* is the name of new symbol, and *ilist* is a comma-separated list of indices types. For example, Gell-Mann matrices can be defined as (although color group and its indices types are already defined):

```
special lambda:(color c3, color c3b, color c8).
```

To define Dirac's γ -matrices we can use the command

```
special gamma:(spinor, cspinor, vector).
```

¹The exception is Lorentz group, corresponding indices types are defined by single keyword.

7.3 Arrays

Array, i.e. the object with explicit components, can also have the user-defined type of index. In this case generic form of such object is

$$\{ \text{expr1}, \text{expr2} \dots, \text{exprN} ; \text{itype} \}$$

where N expressions $\text{expr1} \dots \text{exprN}$ of N components are separated by comma, and itype is an optional index type. If itype is omitted LanHEP uses default group name `wild` and index type `wild N`, where N is a number of components in the array.

8 Auxiliary statements

8.1 Orthogonal matrices

If some parameters appear to be the elements of the orthogonal matrix such as quark mixing Cabbibo-Kobayashi-Maskawa matrix, one should declare them by the statement

$$\text{OrthMatrix}(\{ \{a_{11}, a_{12}, a_{13}\}, \{a_{21}, a_{22}, a_{23}\}, \{a_{31}, a_{32}, a_{33}\} \}).$$

where a_{ij} denote the parameters. Such declaration permits LanHEP to reduce expressions which contain these parameters by taking into account the properties of orthogonal matrices.

Note that this statement has no relation to the arrays; it just declares that these parameters a_{ij} satisfy the correspondent relations. Of course, one can declare further a matrix with these parameters as components by means of `let` statement.

8.2 Including files

LanHEP allows the user to divide the input file into several files. To include the file *file*, the user should use the statement

`read file.`

The standard extension `.mdl` of the file name may be omitted in this statement.

Another way to include a file is provided by the `use` statement as

`use file.`

The `use` statement reads the `file` only once, next appearances of this statement with the same argument do nothing. This function prevents multiple reading of the same file. This form can be used mainly to include some standard modules, such as declaration of Standard Model particles to be used for writing some extensions of this model.

8.3 Checking electric charge conservation

LanHEP can check whether the introduced vertices satisfy electric charge conservation law. This option is available, if the user declares some parameter to denote elementary electric charge (say, `ee` in QED example), and then indicate, which particle is a photon by the statement

$$\text{SetEM}(\text{photon}, \text{param}).$$

So, in example of Section 1 this statement could be

$$\text{SetEM}(\text{A}, \text{ee}).$$

Electric charge of each particle is determined by analyzing its interaction with the photon. LanHEP checks whether the sum of electric charges of particles in each vertex equals zero.

9 LaTeX output

LanHEP generates LaTeX output instead of CompHEP model files if user set `-tex` in the command line to start LanHEP. Three files are produced: `'varsN.tex'`, `'prtclsN.tex'` and `'lgrngN.tex'`. The first file contains names of parameter used in physical model and their values. The second file describes introduced particles, together with propagators derived from introduced vertices.

The last file lists introduced vertices. LanHEP uses Greece letters $\mu, \nu, \rho \dots$ for vector indices, letters $a, b, c \dots$ for spinor ones and $p, q, r \dots$ for color indices (and for indices of other groups, if they were defined).

It is possible to inscribe names for particles and parameters to use them in LaTeX output. It can be done by the statement

```
SetTexName([ ident=texname, ... ]).
```

Here *ident* is an identifier of particle or parameter, and *texname* is string constant containing LaTeX command. Note, that for introducing backslash `'\'` in quoted string constant one should type it twice: `'\\'`.

For example, if one has declared neutrino with name `n1` (and name for antineutrino `N1`) than the statement

```
SetTexName([n1='\\nu^e', N1='\\bar{\\nu}^e']).
```

makes LanHEP to use symbols ν^e and $\bar{\nu}^e$ for neutrino and antineutrino in LaTeX tables.

10 Running LanHEP

As it was mentioned above, LanHEP can read the model description from the input file prepared by user. To start LanHEP write the command

```
lhép filename options
```

where the possible *options* are described in the next section. If the *filename* is omitted, LanHEP prints it's prompt and waits for the keyboard input. In the last case, user's input is copied into the file `lhép.log` and can be inspected in the following. To finish the work with LanHEP, type `'quit.'` or simply press `^D` (or `^Z` at MS DOS computers).

10.1 Options

Possible options, which can be used in the command line to start LanHEP are:

`-OutDir directory` Set the directory where output files will be placed.

`-InDir directory` Set the default directory where to search files, which included by `read` and `use` statements.

`-tex` LanHEP generates LaTeX files instead of CompHEP model tables.

`-frc` If `-tex` option is set, forces LanHEP to split 4-fermion and 4-color vertices just as it is made for CompHEP files.

`-texLines num` Set number of lines in LaTeX tables to *num*. After the specified number of lines, LanHEP continues writing current table on the next page of LaTeX the output. Default value is 40.

`-texLineLength num` Controls width of the Lagrangian table. Default value is 35, user can vary table width by changing this parameter.

11 Default objects

When LanHEP starts, it has already declared some frequently used symbols. They are:

- specials `gamma` and `gamma5` which are Dirac's γ -matrices;
- special `moment` is replaced by the momentum of the corresponding particle;
- let-substitution `deriv` is defined as `-i*moment`;
- group `color` with indices types `c3/c3b` and `c8`;

- specials `lambda` and `f_SU3` which are Gell-Mann matrices and structure constant of $SU(3)$ group;
- `tau1`, `tau2`, `tau3` are τ -matrices τ_1, τ_2, τ_3 ;
- `tau` is a vector $\vec{\tau} = (\tau_1, \tau_2, \tau_3)$;
- `taup` and `taum` are matrices $\tau^\pm = (\tau^1 \pm i\tau^2)/\sqrt{2}$;
- `taupm` is a vector (τ^+, τ^3, τ^-) ;
- `eps` is the antisymmetrical tensor ε ($\varepsilon^{123} = 1$);
- `Tau1`, `Tau2`, `Tau3` are generators of $SU(2)$ group adjoint representation (3-dimensional analog of τ -matrices) T^1, T^2, T^3 with commutative relations $[T_i, T_j] = -i\epsilon_{ijk}T_k$:

$$T^1 = \frac{1}{\sqrt{2}} \begin{pmatrix} 0 & -1 & 0 \\ -1 & 0 & 1 \\ 0 & 1 & 0 \end{pmatrix}, \quad T^2 = \frac{1}{\sqrt{2}} \begin{pmatrix} 0 & i & 0 \\ -i & 0 & -i \\ 0 & i & 0 \end{pmatrix}, \quad T^3 = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & -1 \end{pmatrix};$$

- `Taup` and `Taum` corresponds to $T^\pm = (T^1 \pm iT^2)/\sqrt{2}$;
- `Taupm` is a vector $\vec{T} = (T^+, T^3, T^-)$.

12 Standard Model

In this chapter we give an example of the LanHEP input file introducing the Lagrangian of the Standard Model in the t'Hooft-Feynman gauge.

We start the description of the SM Lagrangian from the declaration of all parameters, see Fig. 3. The comments which follow each parameter explain its meaning.

```
model 'Standard Model'/4.
parameter
  EE = 0.31333:'Elementary electric charge',
  GG = 1.117:'Strong coupling constant (Z point) (PDG-94)',
  SW = 0.4740:'sin of the Weinberg angle (PDG-94,"on-shell")',
  s12 = 0.221:'Parameter of C-K-M matrix (PDG-94)',
  s23 = 0.040:'Parameter of C-K-M matrix (PDG-94)',
  s13 = 0.0035:'Parameter of C-K-M matrix (PDG-94)'.
parameter CW = sqrt(1-SW**2) : 'cos of the Weinberg angle'.
parameter
  c12 = sqrt(1-s12**2):'parameter of C-K-M matrix',
  c23 = sqrt(1-s23**2):'parameter of C-K-M matrix',
  c13 = sqrt(1-s13**2):'parameter of C-K-M matrix'.
parameter
  Vud = c12*c13:'C-K-M matrix element',
  Vus = s12*c13:'C-K-M matrix element',
  Vub = s13:'C-K-M matrix element',
  Vcd = (-s12*c23-c12*s23*s13):'C-K-M matrix element',
  Vcs = (c12*c23-s12*s23*s13):'C-K-M matrix element',
  Vcb = s23*c13:'C-K-M matrix element',
  Vtd = (s12*s23-c12*c23*s13):'C-K-M matrix element',
  Vts = (-c12*s23-s12*c23*s13):'C-K-M matrix element',
  Vtb = c23*c13:'C-K-M matrix element'.
OrthMatrix( {{Vud,Vus,Vub}, {Vcd,Vcs,Vcb}, {Vtd,Vts,Vtb}} ).
```

Figure 3: Standard Model: parameters.

Then we go to input particles of the model. These are vector gauge bosons, fermions which are leptons and quarks of three generations, and scalar Higgs boson (see Fig. 4).

The next section of the input file (Fig. 5) introduces some useful let-substitutions. First, we enter doublets of leptons and quarks; for the quarks we supply also doublets rotated by CKM matrix.

In the textbooks the physical gauge bosons W^\pm , Z , A are defined through the $SU(2)$ gauge triplet W^a ($a = 1, 2, 3$) and $U(1)$ field B as

$$\begin{aligned} W^\pm &= (W^1 \mp iW^2)/\sqrt{2}, \\ Z &= W^3 \cos \theta_W - B \sin \theta_W, \\ A &= W^3 \sin \theta_W + B \cos \theta_W, \end{aligned}$$

where θ_W is the Weinberg angle. If we would like to use the fields W^a and B we have to declare them (as let-substitutions) through the physical fields W^\pm , Z and A by inverting these relations². At last, we define EW charges $g = e/\sin \theta_W$ and $g_1 = e/\cos \theta_W$.

Now we can enter the Lagrangian terms. First, we define the EW gauge bosons Lagrangian (see Fig. 6) in the form

$$L_G^{EW} = -\frac{1}{4}F^{\mu\nu}F_{\mu\nu} - \frac{1}{4}W^{a\mu\nu}W_{\mu\nu}^a,$$

where

$$F_{\mu\nu} = \partial_\mu B_\nu - \partial_\nu B_\mu \text{ and } W_{\mu\nu}^a = \partial_\mu W_\nu^a - \partial_\nu W_\mu^a - g\varepsilon^{abc}W_\mu^b W_\nu^c,$$

and for the gluon Lagrangian we use the form

$$L_G^{QCD} = -\frac{1}{4}G^{a\mu\nu}G_{\mu\nu}^a,$$

²Note, that **B1** stands for the B gauge field since **B** is already defined as b-antiquark.


```

vector
  A/A: (photon, gauge),
  Z/Z: ('Z boson', mass MZ = 91.187, width wZ = 2.502, gauge),
  G/G: (gluon, color c8, gauge),
  'W+'/'W-': ('W boson', mass MW = MZ*CW, width wW = 2.094, gauge).
spinor
  n1:(neutrino,left), e1:(electron, mass Me = 0.000511),
  n2:('mu-neutrino',left), e2:(muon, mass Mm = 0.1057),
  n3:('tau-neutrino',left), e3:('tau-lepton', mass Mt = 1.777).
spinor
  u:('u-quark',color c3),
  d:('d-quark',color c3),
  c:('c-quark',color c3, mass Mc = 1.300),
  s:('s-quark',color c3, mass Ms = 0.200),
  t:('t-quark',color c3, mass Mtop = 170, width wtop = 1.442),
  b:('b-quark',color c3, mass Mb = 4.300 ).
scalar H/H:(Higgs, mass MH = 200, width wH = 1.461).

```

Figure 4: Standard Model: particles.

```

let l1={n1,e1}, L1={N1,E1}.
let l2={n2,e2}, L2={N2,E2}.
let l3={n3,e3}, L3={N3,E3}.
let q1={u,d}, Q1={U,D}, q1a={u,Vud*d+Vus*s+Vub*b}, Q1a={U,Vud*D+Vus*S+Vub*B}.
let q2={c,s}, Q2={C,S}, q2a={c,Vcd*d+Vcs*s+Vcb*b}, Q2a={C,Vcd*D+Vcs*S+Vcb*B}.
let q3={t,b}, Q3={T,B}, q3a={t,Vtd*d+Vts*s+Vtb*b}, Q3a={T,Vtd*D+Vts*S+Vtb*B}.
let B1= -SW*Z+CW*A, W3=CW*Z+SW*A, W1=('W+'+'W-')/Sqrt2, W2 = i*('W+'-'W-')/Sqrt2.
let WW={'W+', W3, 'W-'}, WWc={'W-', W3, 'W+'}, WW1={W1, W2, W3}.
let gh1=('W+.c'+'W-.c')/Sqrt2, gh2= i*('W+.c'-'W-.c')/Sqrt2, gh3= CW*'Z.c'+SW*'A.c',
  gh={gh1,gh2,gh3}.
let Gh1 = ('W+.C'+'W-.C')/Sqrt2, Gh2=i*('W+.C'-'W-.C')/Sqrt2, Gh3= CW*'Z.C'+SW*'A.C',
  Gh={Gh1,Gh2,Gh3}.
let g=EE/SW, g1=EE/CW.

```

Figure 5: Standard Model: useful substitutions.

where

$$G_{\mu\nu}^a = \partial_\mu G_\nu^a - \partial_\nu G_\mu^a - g_s f^{abc} G_\mu^b G_\nu^c,$$

G_μ^a is the gluon field, g_s is a strong interaction coupling.

```

let tB^mu^nu=deriv^mu*B1^nu-deriv^nu*B1^mu.
let tW^mu^nu^a=deriv^mu*WW1^nu^a - deriv^nu*WW1^mu^a - g*eps^a^b^c*WW1^mu^b*WW1^nu^c.
let tG^mu^nu^a=deriv^mu*G^nu^a - deriv^nu*G^mu^a - i*GG*f^a^b^c*G^mu^b*G^nu^c.
lterm -tB**2/4-tW**2/4-tG**2/4.

```

Figure 6: Standard Model: gauge field Lagrangians

The next step is the interaction of fermions with gauge bosons (Fig. 7). First, we define interaction of left-handed fermions with W^a and B fields, taking into account CKM mixing of quarks; the corresponding term is

$$\bar{\psi} \gamma^\mu \frac{1-\gamma^5}{2} \left(i \partial_\mu - \frac{g}{2} \tau^a W_\mu^a - g_1 Y B_\mu \right) \psi,$$

where ψ are left-handed lepton and rotated by CKM matrix quark doublets, Y is a hypercharge. Than we declare the right-hand fermions interaction with B field; in this case we can take quarks without CKM mixing,

since the interaction of quarks with Z -boson and photon is flavor-diagonal. Lagrangian term is

$$\bar{\psi}\gamma^\mu\frac{1+\gamma^5}{2}(i\partial_\mu - g_1 Y B_\mu)\psi,$$

where ψ are right-handed lepton and quark singlets.

```
lterm anti(psi)*gamma*(1-g5)/2*(i*deriv-g*taupm*WW/2-Y*g1*B1)*psi      where
  psi=l1, Y= -1/2;
  psi=l2, Y= -1/2;
  psi=l3, Y= -1/2;
  psi=q1, Y=1/6;
  psi=q2, Y=1/6;
  psi=q3, Y=1/6.
lterm anti(psi)*gamma*(1+g5)/2*(i*deriv-Y*g1*B1)*psi
  where
  psi=e1, Y= -1;
  psi=e2, Y= -1;
  psi=e3, Y= -1;
  psi=u, Y=2/3;
  psi=c, Y=2/3;
  psi=t, Y=2/3;
  psi=d, Y=-1/3;
  psi=s, Y=-1/3;
  psi=b, Y=-1/3.
```

Figure 7: Standard Model: gauge interaction of fermions.

The next step is the introduction of the Higgs sector: interaction of Higgs doublet expressed through physical field H and unphysical goldstone fields Z_f, W_f^\pm as

$$\Phi = \begin{pmatrix} -iW_f^+ \\ \frac{1}{\sqrt{2}}(v + H + iZ_f) \end{pmatrix},$$

where vacuum expectation value is $v = 2M_W/g$. Unphysical goldstone fields read in LanHEP as 'W+.f', 'W-.f', 'Z.f'. In Fig. 8 we enter self-interaction terms and interaction of this field with gauge bosons as:

$$L_{Higgs} = (D_\mu \Phi)^*(D^\mu \Phi) + 2\lambda(\Phi^* \Phi - v^2/2)^2,$$

where $D_\mu = \partial_\mu + i\frac{g}{2}\tau^a W_\mu^a + i\frac{g_1}{2}B_\mu$, $\lambda = \frac{g^2 M_H^2}{16M_W^2}$.

```
let phi = { -i*'W+.f', (vev(2*MW/EE*SW)+H+i*'Z.f')/Sqrt2 },
  Phi = { i*'W-.f', (vev(2*MW/EE*SW)+H-i*'Z.f')/Sqrt2 }.
lterm -2*lambda*(pp*PP+v**2/2)**2
  where lambda=(EE*MH/(MW*SW))*2/16, v=2*MW*SW/EE.
lterm (DPhi)*(DPhi) where
  DPhi = (deriv^mu-i*g1/2*B1^mu)*Phi^a - i*g/2*taupm^a^b^c*WWc^mu^c*Phi^b,
  DPhi = (deriv^mu+i*g1/2*B1^mu)*Phi^a + i*g/2*taupm^a^b^c*WWc^mu^c*Phi^b.
```

Figure 8: Standard Model: self-interaction of Higgs field and its interaction with gauge bosons.

Yukawa terms are shown in Fig. 9. First, we generate lepton masses with terms

$$L_Y = -\frac{gM_\ell}{\sqrt{2}M_W} \left(\bar{\psi}^a \frac{1+\gamma^5}{2} \psi \Phi^a + \bar{\psi} \frac{1-\gamma^5}{2} \psi^a \Phi^{*a} \right),$$

where ψ^a is lepton doublet, ψ is down lepton and M_ℓ is its mass. The term is applied to the leptons of second and third generations (electron is massless in this model).

Next we do the same thing with down quarks. Here we must take into account the CKM mixing, so the previous formula takes the form

$$L_Y^{Q_d} = -\frac{gM_q^{ij}}{\sqrt{2}M_W} \left(\bar{\psi}_i^a \frac{1+\gamma^5}{2} \psi_j \Phi^a + \bar{\psi}_i \frac{1-\gamma^5}{2} \psi_j^a \Phi^{*a} \right),$$

where $M_q^{ij} = \text{diag}(M_d, M_s, M_b)^{ik} U^{jk}$, U being the CKM matrix; ψ_i^a and ψ_i are quark doublets and down singlets, indices i, j numerate generations.

At last we generate masses of upper quarks with the terms

$$L_Y^{Q_u} = -\frac{gM_q}{\sqrt{2}M_W} \left(\bar{\psi}^a \frac{1+\gamma^5}{2} i\tau_2^{ab} \psi \Phi^{*b} + \bar{\psi} \frac{1-\gamma^5}{2} \psi^a i\tau_2^{ab} \Phi^b \right)$$

for three generations of quarks with $M_q = (M_u, M_c, M_t)$.

```
lterm -M/MW/Sqrt2*g*(anti(pl)*(1+g5)/2*pr*phi + anti(pr)*(1-g5)/2*pl*Phi ) where
      M=Mm, pl=l2, pr=e2;
      M=Mt, pl=l3, pr=e3.
lterm -M/MW/Sqrt2*g*(anti(pl)*(1+g5)/2*pr*phi + anti(pr)*(1-g5)/2*pl*Phi ) where
      M=Vud*0, pl=q1a, pr=d; % here and below 0 stands for Md
      M=Vus*Ms, pl=q1a, pr=s;
      M=Vub*Mb, pl=q1a, pr=b;
      M=Vcd*0, pl=q2a, pr=d;
      M=Vcs*Ms, pl=q2a, pr=s;
      M=Vcb*Mb, pl=q2a, pr=b;
      M=Vtd*0, pl=q3a, pr=d;
      M=Vts*Ms, pl=q3a, pr=s;
      M=Vtb*Mb, pl=q3a, pr=b.
lterm -M/MW/Sqrt2*g*(anti(pl)*(1+g5)/2*i*tau2*pr*Phi + anti(pr)*(1-g5)/2*i*pl*tau2*phi )
      where
      M=0, pl=q1a, pr=u;
      M=Mc, pl=q2a, pr=c;
      M=Mtop, pl=q3a, pr=t.
```

Figure 9: Standard Model: Yukawa terms.

The final stage is the introduction of gauge fixing terms and terms with Faddeev-Popov ghosts.

Gauge fixing terms in the t'Hooft-Feynman gauge read as

$$L_{GF} = -\frac{1}{2}(\partial_\mu A^\mu)^2 - \frac{1}{2}(\partial_\mu G_a^\mu)^2 - (\partial_\mu W^{+\mu} + M_W W_f^+)(\partial_\mu W^{-\mu} + M_W W_f^-) - \frac{1}{2}(\partial_\mu Z^\mu + M_Z Z_f)^2,$$

the corresponding LanHEP code is shown in Fig. 10.

```
lterm -1/2*(deriv*A)**2.
lterm -1/2*(deriv*G)**2.
lterm -1/2*(2*(deriv*'W+'MW*'W+.f')*(deriv*'W-'MW*'W-.f') +
      (deriv*Z+MW/CW*'Z.f')**2).
```

Figure 10: Standard Model: gauge fixing terms.

The interaction of ghost fields reads as (we omit here bilinear terms, since they are not treated by CompHEP):

$$L_{FP}^G = i g_s f^{abc} \bar{C}_G^a G_\mu^b \partial^\mu C_G^c$$

for gluons, and

$$L_{FP}^{EW} = -g \varepsilon^{abc} * C_W^a W_\mu^b \bar{C}_W^c$$

for the interaction of EW ghosts with gauge bosons, and finally

$$L_{FP}^{GS} = -\frac{eM_W}{2\sin\theta_W} ((H + iZ_f)(\bar{C}_- C_+ + \bar{C}_+ C_-) +$$

$$\begin{aligned}
& +H\bar{C}_Z C_Z / \cos^2 \theta_W - 2iZ_f \bar{C}_+ C_-) + \\
& + \frac{ieM_W}{2 \cos \theta_W \sin \theta_W} (W_f^+ ((1 - 2 \sin^2 \theta_W) \bar{C}_- C_Z + C_- \bar{C}_Z + 2 \sin \theta_W \cos \theta_W \bar{C}_- C_A) - \\
& - W_f^- ((1 - 2 \sin^2 \theta_W) \bar{C}_+ C_Z + C_+ \bar{C}_Z + 2 \sin \theta_W \cos \theta_W \bar{C}_+ C_A)),
\end{aligned}$$

Here C_G^a (a=1..8) are ghost fields corresponding to gluon, C_W^a (a=1..3) are ghost fields of $SU(2)$ gauge bosons, C_B corresponds to $U(1)$ gauge field and

$$\begin{aligned}
C_{\pm} &= (C_W^1 \mp C_W^2) / \sqrt{2}, \\
C_Z &= C_W^3 \cos \theta_W - C_B \sin \theta_W, \\
C_A &= C_W^3 \sin \theta_W + C_B \cos \theta_W.
\end{aligned}$$

The corresponding part of LanHEP input file is shown in Fig. 11.

```

lterm i*GG*f*ccghost(G)*G*deriv*ghost(G).
lterm i*g*eps*Gh*WW*deriv*gh.
lterm -1/2*(2*(deriv*'W++MW*'W+.f')*(deriv*'W-+MW*'W-.f') +
      (deriv*Z+MW/CW*'Z.f')**2).
lterm -MW*EE/2/SW*((H+i*'Z.f')*( 'W-.C''*W+.c' + 'W+.C''*W-.c')
      +H*'Z.C''*Z.c'/CW**2-2*i*'Z.f'*'W+.C''*W-.c')). lterm
i*EE*MW/2/CW/SW*( 'W+.f'*( 'W-.C''*Z.c'*(1-2*SW**2)+ 'W-.c''*Z.C'
      +2*CW*SW*'W-.C''*A.c') -
      'W-.f'*( 'W+.C''*Z.c'*(1-2*SW**2)+ 'W+.c''*Z.C'
      +2*CW*SW*'W+.C''*A.c')).

```

Figure 11: Standard Model: terms with Faddeev-Popov ghost fields.

To complete the description of Standard Model, we prescribe also LaTeX names for particles and parameters, see Fig. 12.

```

SetTexName([u=u, U='\\bar{u}', d=d, D='\\bar{d}']).
SetTexName([s=s, S='\\bar{s}', b=b, B='\\bar{b}']).
SetTexName([c=c, C='\\bar{c}', t=t, T='\\bar{t}']).
SetTexName([e1=e, E1='\\bar{e}', n1='\\nu^e', N1='\\bar{\\nu}^e']).
SetTexName([e2='\\mu', E2='\\bar{\\mu}', n2='\\nu^\\mu', N2='\\bar{\\nu}^\\mu']).
SetTexName([e3='\\tau', E3='\\bar{\\tau}', n3='\\nu^\\tau', N3='\\bar{\\nu}^\\tau']).
SetTexName([EE=e, GG='g_s', SW='s_w', CW='c_w', MZ='M_Z']).
SetTexName([Me='M_e', Mm='M_\\mu', Mt='M_\\tau', Ms='M_s', Mc='M_c', Mb='M_b', Mtop='M_t']).

```

Figure 12: Standard Model: LaTeX names for particles and parameters.

Appendix


A Processing 4-color vertices

CompHEP Lagrangian tables don't describe explicitly color structure of a vertex. If color particles present in the vertex, the following implicit convolutions are assumed (supposing p, q, r are color indices of particles in the vertex):

- δ_{pq} for two color particles A_p^1, A_q^2 ;
- λ_{pq}^r for three particles, which are color triplet, antitriplet and octet;
- f^{pqr} for three color octets.

Other color structures are forbidden in CompHEP.

So, to introduce the 4-gluon vertex $f^{pqr} G_\mu^q G_\nu^r f^{pst} G_\mu^s G_\nu^t$ one should split this 4-legs vertex into 3-legs vertices $f^{pqr} G_\mu^q G_\nu^r X_{\mu\nu}^p$:



Here the field $X_{\mu\nu}^p$ is Lorentz tensor and color octet, and this field also has constant propagator. If gluon name in CompHEP is 'G', one can use name 'G.t' for this tensor particle; its indices denoted as 'm_' and 'M_' ('_' is the number of the particle in table item).

B Feynman rules for the Standard Model

Here we present the tables generated by LanHEP in LaTeX format for the Standard Model (see Section 12).

Table 2: Standard model: particles.

P	aP	Name	Spin	EM charge	Color	2-legs vertex	Comment
A_μ	A_ν	photon	1	0	1	$-p_1^\rho p_1^\rho g^{\mu\nu}$	gauge
Z_μ	Z_ν	Z boson	1	0	1	$-\frac{1}{c_w^2} (c_w^2 p_1^\rho p_1^\rho g^{\mu\nu} - MW^2 \cdot g^{\mu\nu})$	gauge
$G_{\mu p}$	$G_{\nu q}$	gluon	1	0	8	$-p_1^\rho p_1^\rho g^{\mu\nu} \delta_{pq}$	gauge
W_μ^+	W_ν^-	W boson	1	1	1	$-g^{\mu\nu} (p_1^\rho p_1^\rho - MW^2)$	gauge
ν_a^e	$\bar{\nu}_b^e$	neutrino	1/2	0	1	$-\frac{1}{2} p_1^\mu (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$	left
e_a	\bar{e}_b	electron	1/2	-1	1	$-p_1^\mu \gamma_{ab}^\mu$	
ν_a^μ	$\bar{\nu}_b^\mu$	mu-neutrino	1/2	0	1	$-\frac{1}{2} p_1^\mu (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$	left
μ_a	$\bar{\mu}_b$	muon	1/2	-1	1	$-(p_1^\mu \gamma_{ab}^\mu + M_\mu \cdot \delta_{ab})$	
ν_a^τ	$\bar{\nu}_b^\tau$	tau-neutrino	1/2	0	1	$-\frac{1}{2} p_1^\mu (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$	left
τ_a	$\bar{\tau}_b$	tau-lepton	1/2	-1	1	$-(p_1^\mu \gamma_{ab}^\mu + M_\tau \cdot \delta_{ab})$	
u_{ap}	\bar{u}_{bq}	u-quark	1/2	2/3	3	$-p_1^\mu \delta_{pq} \gamma_{ab}^\mu$	
d_{ap}	\bar{d}_{bq}	d-quark	1/2	-1/3	3	$-\frac{1}{2} p_1^\mu \delta_{pq} 2\gamma_{ab}^\mu$	
c_{ap}	\bar{c}_{bq}	c-quark	1/2	2/3	3	$-\delta_{pq} (p_1^\mu \gamma_{ab}^\mu + M_c \cdot \delta_{ab})$	
s_{ap}	\bar{s}_{bq}	s-quark	1/2	-1/3	3	$-\frac{1}{2} \delta_{pq} (2p_1^\mu \gamma_{ab}^\mu + 2M_s \cdot \delta_{ab})$	
t_{ap}	\bar{t}_{bq}	t-quark	1/2	2/3	3	$-\delta_{pq} (p_1^\mu \gamma_{ab}^\mu + M_t \cdot \delta_{ab})$	
b_{ap}	\bar{b}_{bq}	b-quark	1/2	-1/3	3	$-\frac{1}{2} \delta_{pq} (2p_1^\mu \gamma_{ab}^\mu + 2M_b \cdot \delta_{ab})$	
H	H	Higgs	0	0	1	$-(MH^2 - p_1^\mu p_1^\mu)$	

Table 3: Standard Model: parameters.

Parameter	Value	Comment
EE	0.31333	Electromagnetic coupling constant (1/128)
GG	1.117	Strong coupling constant (Z point) (PDG-94)
SW	0.474	sin of the Weinberg angle (PDG-94,"on-shell")
s12	0.221	Parameter of C-K-M matrix (PDG-94)
s23	0.04	Parameter of C-K-M matrix (PDG-94)
s13	0.0035	Parameter of C-K-M matrix (PDG-94)
CW	$\text{Sqrt}(1-\text{SW}^2)$	cos of the Weinberg angle
c12	$\text{Sqrt}(1-s12^2)$	parameter of C-K-M matrix
c23	$\text{Sqrt}(1-s23^2)$	parameter of C-K-M matrix
c13	$\text{Sqrt}(1-s13^2)$	parameter of C-K-M matrix
Vud	$c12*c13$	C-K-M matrix element
Vus	$s12*c13$	C-K-M matrix element
Vub	$s13$	C-K-M matrix element
Vcd	$-s12*c23-c12*s23*s13$	C-K-M matrix element
Vcs	$c12*c23-s12*s23*s13$	C-K-M matrix element
Vcb	$s23*c13$	C-K-M matrix element
Vtd	$s12*s23-c12*c23*s13$	C-K-M matrix element
Vts	$-c12*s23-s12*c23*s13$	C-K-M matrix element
Vtb	$c23*c13$	C-K-M matrix element
MZ	91.187	mass of Z boson
wZ	2.502	width of Z boson
MW	$MZ*CW$	mass of W boson
wW	2.094	width of W boson
Me	0.000511	mass of electron
Mm	0.1057	mass of muon
Mt	1.777	mass of tau-lepton
Mc	1.3	mass of c-quark
Ms	0.2	mass of s-quark
Mtop	170	mass of t-quark
wtop	1.442	width of t-quark
Mb	4.3	mass of b-quark
MH	200	mass of Higgs
wH	1.461	width of Higgs

Fields in the vertex				Variational derivative of Lagrangian by fields
$G_{\mu p}$	$G_{\nu q}$	$G_{\rho r}$		$g_s f_{pqr} (p_3^\nu g^{\mu\rho} - p_2^\rho g^{\mu\nu} - p_3^\mu g^{\nu\rho} + p_1^\rho g^{\mu\nu} + p_2^\mu g^{\nu\rho} - p_1^\nu g^{\mu\rho})$
$G_{\mu p}$	$G_{\nu q}$	$G_{\rho r}$	$G_{\sigma s}$	$g_s^2 (g^{\mu\rho} g^{\nu\sigma} f_{pqt} f_{rst} - g^{\mu\sigma} g^{\nu\rho} f_{pqt} f_{rst} + g^{\mu\nu} g^{\rho\sigma} f_{prt} f_{qst}$ $+ g^{\mu\nu} g^{\rho\sigma} f_{pst} f_{qrt} - g^{\mu\sigma} g^{\nu\rho} f_{prt} f_{qst} - g^{\mu\rho} g^{\nu\sigma} f_{pst} f_{qrt})$ $- \frac{c_w e}{s_w} (p_1^\nu g^{\mu\rho} - p_1^\rho g^{\mu\nu} - p_2^\mu g^{\nu\rho} + p_2^\rho g^{\mu\nu} + p_3^\mu g^{\nu\rho} - p_3^\nu g^{\mu\rho})$ $- e (p_2^\rho g^{\mu\nu} - p_2^\mu g^{\nu\rho} - p_3^\nu g^{\mu\rho} + p_3^\mu g^{\nu\rho} + p_1^\nu g^{\mu\rho} - p_1^\rho g^{\mu\nu})$ $- \frac{c_w e^2}{s_w^2} (2g^{\mu\nu} g^{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma} - g^{\mu\sigma} g^{\nu\rho})$ $- \frac{c_w e^2}{s_w} (2g^{\mu\sigma} g^{\nu\rho} - g^{\mu\nu} g^{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma})$ $- e^2 (2g^{\mu\nu} g^{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma} - g^{\mu\sigma} g^{\nu\rho})$ $\frac{e^2}{s_w^2} (2g^{\mu\nu} g^{\rho\sigma} - g^{\mu\sigma} g^{\nu\rho} - g^{\mu\rho} g^{\nu\sigma})$
W^+_μ	W^-_ν	Z_ρ		$-\frac{c_w e}{s_w} (p_1^\nu g^{\mu\rho} - p_1^\rho g^{\mu\nu} - p_2^\mu g^{\nu\rho} + p_2^\rho g^{\mu\nu} + p_3^\mu g^{\nu\rho} - p_3^\nu g^{\mu\rho})$
A_μ	W^+_ν	W^-_ρ		$-e (p_2^\rho g^{\mu\nu} - p_2^\mu g^{\nu\rho} - p_3^\nu g^{\mu\rho} + p_3^\mu g^{\nu\rho} + p_1^\nu g^{\mu\rho} - p_1^\rho g^{\mu\nu})$
W^+_μ	W^-_ν	Z_ρ	Z_σ	$-\frac{c_w e^2}{s_w^2} (2g^{\mu\nu} g^{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma} - g^{\mu\sigma} g^{\nu\rho})$
A_μ	W^+_ν	W^-_ρ	Z_σ	$-\frac{c_w e^2}{s_w} (2g^{\mu\sigma} g^{\nu\rho} - g^{\mu\nu} g^{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma})$
A_μ	A_ν	W^+_ρ	W^-_σ	$-e^2 (2g^{\mu\nu} g^{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma} - g^{\mu\sigma} g^{\nu\rho})$
W^+_μ	W^+_ν	W^-_ρ	W^-_σ	$\frac{e^2}{s_w^2} (2g^{\mu\nu} g^{\rho\sigma} - g^{\mu\sigma} g^{\nu\rho} - g^{\mu\rho} g^{\nu\sigma})$
$\bar{\nu}^e_a$	ν^e_b	Z_μ		$-\frac{1}{4} \frac{e}{c_w s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\nu}^e_a$	e_b	W^+_μ		$-\frac{1}{4} \frac{e\sqrt{2}}{s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{e}_a	ν^e_b	W^-_μ		$-\frac{1}{4} \frac{e\sqrt{2}}{s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{e}_a	e_b	Z_μ		$\frac{1}{4} \frac{e}{c_w s_w} (-4s_w^2 \gamma_{ab}^\mu + \gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{e}_a	e_b	A_μ		$e\gamma_{ab}^\mu$
$\bar{\nu}^\mu_a$	ν^μ_b	Z_μ		$-\frac{1}{4} \frac{e}{c_w s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\nu}^\mu_a$	μ_b	W^+_μ		$-\frac{1}{4} \frac{e\sqrt{2}}{s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\mu}_a$	ν^μ_b	W^-_μ		$-\frac{1}{4} \frac{e\sqrt{2}}{s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\mu}_a$	μ_b	Z_μ		$\frac{1}{4} \frac{e}{c_w s_w} (-4s_w^2 \gamma_{ab}^\mu + \gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\mu}_a$	μ_b	A_μ		$e\gamma_{ab}^\mu$
$\bar{\nu}^\tau_a$	ν^τ_b	Z_μ		$-\frac{1}{4} \frac{e}{c_w s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\nu}^\tau_a$	τ_b	W^+_μ		$-\frac{1}{4} \frac{e\sqrt{2}}{s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\tau}_a$	ν^τ_b	W^-_μ		$-\frac{1}{4} \frac{e\sqrt{2}}{s_w} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\tau}_a$	τ_b	Z_μ		$\frac{1}{4} \frac{e}{c_w s_w} (-4s_w^2 \gamma_{ab}^\mu + \gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
$\bar{\tau}_a$	τ_b	A_μ		$e\gamma_{ab}^\mu$
\bar{u}_{ap}	u_{bq}	Z_μ		$-\frac{1}{12} \frac{e}{c_w s_w} \delta_{pq} (-8s_w^2 \gamma_{ab}^\mu + 3\gamma_{ab}^\mu - 3\gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{u}_{ap}	u_{bq}	A_μ		$-\frac{2}{3} e \delta_{pq} \gamma_{ab}^\mu$
\bar{u}_{ap}	d_{bq}	W^+_μ		$-\frac{1}{4} \frac{e\sqrt{2} V_{ud}}{s_w} \delta_{pq} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{u}_{ap}	s_{bq}	W^+_μ		$-\frac{1}{4} \frac{e\sqrt{2} V_{us}}{s_w} \delta_{pq} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{u}_{ap}	b_{bq}	W^+_μ		$-\frac{1}{4} \frac{e\sqrt{2} V_{ub}}{s_w} \delta_{pq} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{d}_{ap}	u_{bq}	W^-_μ		$-\frac{1}{4} \frac{e\sqrt{2} V_{ud}}{s_w} \delta_{pq} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{d}_{ap}	d_{bq}	Z_μ		$\frac{1}{12} \frac{e}{c_w s_w} \delta_{pq} (-4s_w^2 \gamma_{ab}^\mu + 3\gamma_{ab}^\mu - 3\gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{d}_{ap}	d_{bq}	A_μ		$\frac{1}{6} e \delta_{pq} 2\gamma_{ab}^\mu$
\bar{s}_{ap}	u_{bq}	W^-_μ		$-\frac{1}{4} \frac{e\sqrt{2} V_{us}}{s_w} \delta_{pq} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{s}_{ap}	s_{bq}	Z_μ		$\frac{1}{12} \frac{e}{c_w s_w} \delta_{pq} (-4s_w^2 \gamma_{ab}^\mu + 3\gamma_{ab}^\mu - 3\gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{s}_{ap}	s_{bq}	A_μ		$\frac{1}{6} e \delta_{pq} 2\gamma_{ab}^\mu$
\bar{b}_{ap}	u_{bq}	W^-_μ		$-\frac{1}{4} \frac{e\sqrt{2} V_{ub}}{s_w} \delta_{pq} (\gamma_{ab}^\mu - \gamma_{ac}^\mu \gamma_{cb}^5)$
\bar{b}_{ap}	b_{bq}	Z_μ		$\frac{1}{12} \frac{e}{c_w s_w} \delta_{pq} (-4s_w^2 \gamma_{ab}^\mu + 3\gamma_{ab}^\mu - 3\gamma_{ac}^\mu \gamma_{cb}^5)$

Fields in the vertex			Variational derivative of Lagrangian terms by fields
b_{ap}	b_{bq}	A_μ	$\frac{1}{6}e\delta_{pq}2\gamma_{ab}^\mu$
\bar{c}_{ap}	c_{bq}	Z_μ	$-\frac{1}{12}\frac{e}{c_ws_w}\delta_{pq}(-8s_w^2\gamma_{ab}^\mu+3\gamma_{ab}^\mu-3\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{c}_{ap}	c_{bq}	A_μ	$-\frac{2}{3}e\delta_{pq}\gamma_{ab}^\mu$
\bar{c}_{ap}	d_{bq}	W^+_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vcd}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{c}_{ap}	s_{bq}	W^+_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vcs}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{c}_{ap}	b_{bq}	W^+_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vcb}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{d}_{ap}	c_{bq}	W^-_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vcd}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{s}_{ap}	c_{bq}	W^-_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vcs}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{b}_{ap}	c_{bq}	W^-_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vcb}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{t}_{ap}	t_{bq}	Z_μ	$-\frac{1}{12}\frac{e}{c_ws_w}\delta_{pq}(-8s_w^2\gamma_{ab}^\mu+3\gamma_{ab}^\mu-3\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{t}_{ap}	t_{bq}	A_μ	$-\frac{2}{3}e\delta_{pq}\gamma_{ab}^\mu$
\bar{t}_{ap}	d_{bq}	W^+_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vtd}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{t}_{ap}	s_{bq}	W^+_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vts}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{t}_{ap}	b_{bq}	W^+_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vtb}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{d}_{ap}	t_{bq}	W^-_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vtd}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{s}_{ap}	t_{bq}	W^-_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vts}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{b}_{ap}	t_{bq}	W^-_μ	$-\frac{1}{4}\frac{e\sqrt{2}Vtb}{s_w}\delta_{pq}(\gamma_{ab}^\mu-\gamma_{ac}^\mu\gamma_{cb}^5)$
\bar{u}_{ap}	u_{bq}	$G_{\mu r}$	$g_s\gamma_{ab}^\mu\lambda_{pq}^r$
\bar{d}_{ap}	d_{bq}	$G_{\mu r}$	$g_s\gamma_{ab}^\mu\lambda_{pq}^r$
\bar{c}_{ap}	c_{bq}	$G_{\mu r}$	$g_s\gamma_{ab}^\mu\lambda_{pq}^r$
\bar{s}_{ap}	s_{bq}	$G_{\mu r}$	$g_s\gamma_{ab}^\mu\lambda_{pq}^r$
\bar{t}_{ap}	t_{bq}	$G_{\mu r}$	$g_s\gamma_{ab}^\mu\lambda_{pq}^r$
\bar{b}_{ap}	b_{bq}	$G_{\mu r}$	$g_s\gamma_{ab}^\mu\lambda_{pq}^r$
\bar{u}_{ap}	s_{bq}	W_F^+	$\frac{1}{4}\frac{ieM_s\sqrt{2}Vus}{MWs_w}\delta_{pq}(\delta_{ab}+\gamma_{ab}^5)$
\bar{s}_{ap}	s_{bq}	H	$-\frac{1}{2}\frac{eM_s}{MWs_w}\delta_{ab}\delta_{pq}$
\bar{s}_{ap}	s_{bq}	Z_F	$-\frac{1}{2}\frac{ieM_s}{MWs_w}\delta_{pq}\gamma_{ab}^5$
\bar{s}_{ap}	u_{bq}	W_F^-	$-\frac{1}{4}\frac{ieM_s\sqrt{2}Vus}{MWs_w}\delta_{pq}(\delta_{ab}-\gamma_{ab}^5)$
\bar{u}_{ap}	b_{bq}	W_F^+	$\frac{1}{4}\frac{ieM_b\sqrt{2}Vub}{MWs_w}\delta_{pq}(\delta_{ab}+\gamma_{ab}^5)$
\bar{b}_{ap}	b_{bq}	H	$-\frac{1}{2}\frac{eM_b}{MWs_w}\delta_{ab}\delta_{pq}$
\bar{b}_{ap}	b_{bq}	Z_F	$-\frac{1}{2}\frac{ieM_b}{MWs_w}\delta_{pq}\gamma_{ab}^5$
\bar{b}_{ap}	u_{bq}	W_F^-	$-\frac{1}{4}\frac{ieM_b\sqrt{2}Vub}{MWs_w}\delta_{pq}(\delta_{ab}-\gamma_{ab}^5)$
\bar{c}_{ap}	s_{bq}	W_F^+	$\frac{1}{4}\frac{ie\sqrt{2}Vcs}{MWs_w}\delta_{pq}(M_s\delta_{ab}+M_s\gamma_{ab}^5-M_c\delta_{ab}+M_c\gamma_{ab}^5)$
\bar{s}_{ap}	c_{bq}	W_F^-	$-\frac{1}{4}\frac{ie\sqrt{2}Vcs}{MWs_w}\delta_{pq}(M_s\delta_{ab}-M_s\gamma_{ab}^5-M_c\delta_{ab}-M_c\gamma_{ab}^5)$
\bar{c}_{ap}	b_{bq}	W_F^+	$\frac{1}{4}\frac{ie\sqrt{2}Vcb}{MWs_w}\delta_{pq}(M_b\delta_{ab}+M_b\gamma_{ab}^5-M_c\delta_{ab}+M_c\gamma_{ab}^5)$
\bar{b}_{ap}	c_{bq}	W_F^-	$-\frac{1}{4}\frac{ie\sqrt{2}Vcb}{MWs_w}\delta_{pq}(M_b\delta_{ab}-M_b\gamma_{ab}^5-M_c\delta_{ab}-M_c\gamma_{ab}^5)$
\bar{t}_{ap}	s_{bq}	W_F^+	$\frac{1}{4}\frac{ie\sqrt{2}Vts}{MWs_w}\delta_{pq}(M_s\delta_{ab}+M_s\gamma_{ab}^5-M_t\delta_{ab}+M_t\gamma_{ab}^5)$

Fields in the vertex				Variational derivative of Lagrangian by fields
\bar{s}_{ap}	t_{bq}	W_F^-		$-\frac{1}{4} \frac{ie\sqrt{2}Vts}{MWs_w} \delta_{pq} (M_s \delta_{ab} - M_s \gamma_{ab}^5 - M_t \delta_{ab} - M_t \gamma_{ab}^5)$
\bar{t}_{ap}	b_{bq}	W_F^+		$\frac{1}{4} \frac{ie\sqrt{2}Vtb}{MWs_w} \delta_{pq} (M_b \delta_{ab} + M_b \gamma_{ab}^5 - M_t \delta_{ab} + M_t \gamma_{ab}^5)$
\bar{b}_{ap}	t_{bq}	W_F^-		$-\frac{1}{4} \frac{ie\sqrt{2}Vtb}{MWs_w} \delta_{pq} (M_b \delta_{ab} - M_b \gamma_{ab}^5 - M_t \delta_{ab} - M_t \gamma_{ab}^5)$
\bar{c}_{ap}	c_{bq}	H		$-\frac{1}{2} \frac{eM_c}{MWs_w} \delta_{ab} \delta_{pq}$
\bar{c}_{ap}	c_{bq}	Z_F		$\frac{1}{2} \frac{ieM_c}{MWs_w} \delta_{pq} \gamma_{ab}^5$
\bar{d}_{ap}	c_{bq}	W_F^-		$\frac{1}{4} \frac{ieM_c\sqrt{2}Vcd}{MWs_w} \delta_{pq} (\delta_{ab} + \gamma_{ab}^5)$
\bar{c}_{ap}	d_{bq}	W_F^+		$-\frac{1}{4} \frac{ieM_c\sqrt{2}Vcd}{MWs_w} \delta_{pq} (\delta_{ab} - \gamma_{ab}^5)$
\bar{t}_{ap}	t_{bq}	H		$-\frac{1}{2} \frac{eM_t}{MWs_w} \delta_{ab} \delta_{pq}$
\bar{t}_{ap}	t_{bq}	Z_F		$\frac{1}{2} \frac{ieM_t}{MWs_w} \delta_{pq} \gamma_{ab}^5$
\bar{d}_{ap}	t_{bq}	W_F^-		$\frac{1}{4} \frac{ieM_t\sqrt{2}Vtd}{MWs_w} \delta_{pq} (\delta_{ab} + \gamma_{ab}^5)$
\bar{t}_{ap}	d_{bq}	W_F^+		$-\frac{1}{4} \frac{ieM_t\sqrt{2}Vtd}{MWs_w} \delta_{pq} (\delta_{ab} - \gamma_{ab}^5)$
$\bar{\nu}^\mu_a$	μ_b	W_F^+		$\frac{1}{4} \frac{ieM_\mu\sqrt{2}}{MWs_w} (\delta_{ab} + \gamma_{ab}^5)$
$\bar{\mu}_a$	μ_b	H		$-\frac{1}{2} \frac{eM_\mu}{MWs_w} \delta_{ab}$
$\bar{\mu}_a$	μ_b	Z_F		$-\frac{1}{2} \frac{ieM_\mu}{MWs_w} \gamma_{ab}^5$
$\bar{\mu}_a$	ν^μ_b	W_F^-		$-\frac{1}{4} \frac{ieM_\mu\sqrt{2}}{MWs_w} (\delta_{ab} - \gamma_{ab}^5)$
$\bar{\nu}^\tau_a$	τ_b	W_F^+		$\frac{1}{4} \frac{ieM_\tau\sqrt{2}}{MWs_w} (\delta_{ab} + \gamma_{ab}^5)$
$\bar{\tau}_a$	τ_b	H		$-\frac{1}{2} \frac{eM_\tau}{MWs_w} \delta_{ab}$
$\bar{\tau}_a$	τ_b	Z_F		$-\frac{1}{2} \frac{ieM_\tau}{MWs_w} \gamma_{ab}^5$
$\bar{\tau}_a$	ν^τ_b	W_F^-		$-\frac{1}{4} \frac{ieM_\tau\sqrt{2}}{MWs_w} (\delta_{ab} - \gamma_{ab}^5)$
W_F^+	W_F^+	W_F^-	W_F^-	$-\frac{1}{2} \frac{e^2 MH^2}{MW^2 s_w^2}$
H	W_F^+	W_F^-		$-\frac{1}{2} \frac{eMH^2}{MWs_w}$
H	H	W_F^+	W_F^-	$-\frac{1}{4} \frac{e^2 MH^2}{MW^2 s_w^2}$
W_F^+	W_F^-	Z_F	Z_F	$-\frac{1}{4} \frac{e^2 MH^2}{MW^2 s_w^2}$
H	H	H		$-\frac{3}{2} \frac{eMH^2}{MWs_w}$
H	Z_F	Z_F		$-\frac{1}{2} \frac{eMH^2}{MWs_w}$
H	H	H	H	$-\frac{3}{4} \frac{e^2 MH^2}{MW^2 s_w^2}$
H	H	Z_F	Z_F	$-\frac{1}{4} \frac{e^2 MH^2}{MW^2 s_w^2}$
Z_F	Z_F	Z_F	Z_F	$-\frac{3}{4} \frac{e^2 MH^2}{MW^2 s_w^2}$
W_F^+	W_F^-	Z_μ		$-\frac{1}{2} \frac{e}{c_w s_w} (2s_w^2 p_2^\mu - p_2^\mu - 2s_w^2 p_1^\mu + p_1^\mu)$
H	Z_μ	Z_F		$-\frac{1}{2} \frac{ie}{c_w s_w} (-p_3^\mu + p_1^\mu)$
A_μ	W_F^+	W_F^-		$e(p_3^\mu - p_2^\mu)$
H	W_F^+	W_F^-	W_F^-	$-\frac{1}{2} \frac{ie}{s_w} (p_1^\mu - p_2^\mu)$
W_F^+	W_F^-	Z_F		$-\frac{1}{2} \frac{e}{s_w} (p_3^\mu - p_1^\mu)$
H	W_F^+	W_F^-		$\frac{1}{2} \frac{ie}{s_w} (p_3^\mu - p_1^\mu)$
W_F^+	W_F^-	Z_F		$-\frac{1}{2} \frac{e}{s_w} (p_2^\mu - p_3^\mu)$
W_F^+	W_F^-	Z_μ	Z_ν	$\frac{1}{2} \frac{e^2}{c_w^2 s_w^2} g^{\mu\nu} (4s_w^4 - 4s_w^2 +)$

Fields in the vertex				Variational derivative of Lagrangian by fields
H	Z_μ	Z_ν		$\frac{eMW}{c_w^2 s_w} g^{\mu\nu}$
H	H	Z_μ	Z_ν	$\frac{1}{2} \frac{e^2}{c_w^2 s_w^2} g^{\mu\nu}$
Z_μ	Z_ν	Z_F	Z_F	$\frac{1}{2} \frac{e^2}{c_w^2 s_w^2} g^{\mu\nu}$
A_μ	W_F^+	W_F^-	Z_ν	$-\frac{e^2}{c_w s_w} g^{\mu\nu} (2s_w^2 -)$
W_F^+	W^-_μ	Z_ν		$\frac{ieMW s_w}{c_w} g^{\mu\nu}$
H	W_F^+	W^-_μ	Z_ν	$\frac{1}{2} \frac{ie^2}{c_w} g^{\mu\nu}$
W_F^+	W^-_μ	Z_ν	Z_F	$\frac{1}{2} \frac{e^2}{c_w} g^{\mu\nu}$
W^+_μ	W_F^-	Z_ν		$-\frac{ieMW s_w}{c_w} g^{\mu\nu}$
H	W^+_μ	W_F^-	Z_ν	$-\frac{1}{2} \frac{ie^2}{c_w} g^{\mu\nu}$
W^+_μ	W_F^-	Z_ν	Z_F	$\frac{1}{2} \frac{e^2}{c_w} g^{\mu\nu}$
A_μ	A_ν	W_F^+	W_F^-	$2e^2 g^{\mu\nu}$
A_μ	W_F^+	W^-_ν		$-ieMW g^{\mu\nu}$
A_μ	H	W_F^+	W^-_ν	$-\frac{1}{2} \frac{ie^2}{s_w} g^{\mu\nu}$
A_μ	W_F^+	W^-_ν	Z_F	$-\frac{1}{2} \frac{e^2}{s_w} g^{\mu\nu}$
A_μ	W^+_ν	W_F^-		$ieMW g^{\mu\nu}$
A_μ	H	W^+_ν	W_F^-	$\frac{1}{2} \frac{ie^2}{s_w} g^{\mu\nu}$
A_μ	W^+_ν	W_F^-	Z_F	$-\frac{1}{2} \frac{e^2}{s_w} g^{\mu\nu}$
W^+_μ	W_F^+	W^-_ν	W_F^-	$\frac{1}{2} \frac{e^2}{s_w^2} g^{\mu\nu}$
H	W^+_μ	W^-_ν		$\frac{eMW}{s_w} g^{\mu\nu}$
H	H	W^+_μ	W^-_ν	$\frac{1}{2} \frac{e^2}{s_w^2} g^{\mu\nu}$
W^+_μ	W^-_ν	Z_F	Z_F	$\frac{1}{2} \frac{e^2}{s_w^2} g^{\mu\nu}$
$G_{\mu p}$	\bar{C}^G_q	C^G_r		$g_s P_3^\mu f_{pqr}$
C^{W+}	W^-_μ	\bar{C}^Z		$\frac{c_w e}{s_w} p_3^\mu$
\bar{C}^A	C^{W+}	W^-_μ		ep_1^μ
C^{W+}	\bar{C}^{W-}	Z_μ		$-\frac{c_w e}{s_w} p_2^\mu$
A_μ	C^{W+}	\bar{C}^{W-}		$-ep_3^\mu$
W^+_μ	C^{W-}	\bar{C}^Z		$-\frac{c_w e}{s_w} p_3^\mu$
\bar{C}^A	W^+_μ	C^{W-}		$-ep_1^\mu$
\bar{C}^{W+}	C^{W-}	Z_μ		$\frac{c_w e}{s_w} p_1^\mu$
A_μ	\bar{C}^{W+}	C^{W-}		ep_2^μ
W^+_μ	\bar{C}^{W-}	C^Z		$\frac{c_w e}{s_w} p_2^\mu$
\bar{C}^{W+}	W^-_μ	C^Z		$-\frac{c_w e}{s_w} p_1^\mu$
C^Z	W^+_μ	\bar{C}^{W-}		ep_3^μ
C^Z	\bar{C}^{W+}	W^-_μ		$-ep_2^\mu$
H	C^{W+}	\bar{C}^{W-}		$-\frac{1}{2} \frac{eMW}{s_w}$
H	\bar{C}^{W+}	C^{W-}		$-\frac{1}{2} \frac{eMW}{s_w}$

Fields in the vertex			Variational derivative of Lagrangian by fields
C^{W+}	\bar{C}^{W-}	Z_F	$-\frac{1}{2} \frac{ieMW}{s_w}$
\bar{C}^{W+}	C^{W-}	Z_F	$\frac{1}{2} \frac{ieMW}{s_w}$
H	\bar{C}^Z	C^Z	$-\frac{1}{2} \frac{eMW}{c_w^2 s_w}$
W_F^+	\bar{C}^{W-}	C^Z	$\frac{1}{2} \frac{ieMW}{c_w s_w} (-2s_w^2)$
W_F^+	C^{W-}	\bar{C}^Z	$\frac{1}{2} \frac{ieMW}{c_w s_w}$
C^Z	W_F^+	\bar{C}^{W-}	$ieMW$
\bar{C}^{W+}	W_F^-	C^Z	$-\frac{1}{2} \frac{ieMW}{c_w s_w} (-2s_w^2)$
C^{W+}	W_F^-	\bar{C}^Z	$-\frac{1}{2} \frac{ieMW}{c_w s_w}$
C^Z	\bar{C}^{W+}	W_F^-	$-ieMW$

Acknowledgements

This work was supported by ISSEP grant a96-401, INTAS grant 93-1180-ext, and by the Grant Center for Natural Sciences of State Committee for Higher Education of Russia.

References

- [1] E. Boos et al., INP MSU-94-36/358, SNUTP 94-116 (Seoul, 1994); hep-ph/9503280

Contents

1	QED by means of LanHEP	3
2	QCD	4
3	Syntax	5
3.1	Constants and identifiers	5
3.2	Comments	6
4	Objects in the expressions for Lagrangian terms	6
4.1	Parameters	6
4.2	Particles	7
4.3	Specials	8
4.4	Let-substitutions	8
4.5	Arrays	8
5	Structure of expressions for the Lagrangian	9
5.1	Where-substitutions	9
6	Omitting indices	10
6.1	Restoring the omitted indices	10
6.2	Convolution of restored indices	11
6.3	Let-substitutions	11
6.4	Arrays	11
7	Declaration of new index types and indexed objects	12
7.1	Declaring new groups	12
7.2	Declaring new specials	12
7.3	Arrays	13
8	Auxiliary statements	13
8.1	Orthogonal matrices	13
8.2	Including files	13
8.3	Checking electric charge conservation	13
9	LaTeX output	14
10	Running LanHEP	14
10.1	Options	14
11	Default objects	14
12	Standard Model	16
A	Processing 4-color vertices	21
B	Feynman rules for the Standard Model	21